

LUCRARE DE LABORATOR

SIMULAREA FUNCȚIONĂRII CIRCUITELOR ELECTRONICE CU AJUTORUL PROGRAMULUI PSPICE

1. Prezentare generală

Creșterea complexității circuitelor electronice, performanțele ridicate, necesitatea de a reduce timpul de proiectare și nu în cele din urmă reducerea costurilor de proiectare și realizare a circuitelor integrate au impus în ultimele decenii utilizarea programelor de simulare a circuitelor electronice.

În anii 1970, la Universitatea Berkeley din California, un colectiv de cercetători ce-l includea și pe Andrei Vladimirescu, absolvent al Facultății de Electronică din București, a dezvoltat un program universal de simulare a circuitelor electronice numit *Simulation Program with Integrated Circuit Emphasis* sau pe scurt *SPICE*.

Acest program a fost dezvoltat și realizat ulterior în mai multe variante, de către mai multe firme. Pentru această lucrare de laborator este aleasă varianta PSPICE a firmei MicroSim cu versiunea pentru sistemul de operare DOS.

Programul poate fi utilizat pentru simularea funcționării circuitelor prin efectuarea a 8 tipuri de analiză, după cum urmează:

1. analiza de curent continuu a unei surse de tensiune sau curent de intrare, a unui parametru de model, sau a temperaturii (.DC);
2. analiza punctelor statice de funcționare (.OP), ce face posibilă realizarea următoarelor analize suplimentare:
 3. analiza răspunsului în frecvență (.AC);
 4. analiza de zgomot (.NOISE);
 5. analiza sensibilității de curent continuu (.SENS);
 6. analiza funcției de transfer de semnal mic (.TF);
7. analiza răspunsului tranzitoriu, sau comportarea în timp (.TRAN), ce face posibilă realizarea analizei:
8. analiza componentelor Fourier ale răspunsului tranzitoriu (.FOUR).

Toate aceste analize pot fi efectuate prin inserarea în liniile programului a comenzii dorite, câte una pentru fiecare rulare. Dacă sunt apelate mai multe analize diferite în același program atunci ultima este cea efectuată.

Pentru efectuarea oricărei analize este necesară mai întâi descrierea circuitului. Aceasta constă în:

1. numerotarea nodurilor (un nod fiind punctul din circuit în care se unesc cel puțin 2 terminale ale unor elemente de circuit), nodul de masă fiind obligatoriu notat cu 0 (zero).
2. declararea elementelor de circuit, prin nume, noduri de conectare, valoare și model, delimitate între ele prin spații, astfel:
 - numele este unic iar prima sa literă desemnează tipul dispozitivului, astfel litera R este alocată pentru rezistențe, C pentru condensatoare, Q pentru tranzistoare bipolare, V pentru surse de tensiune etc.;
 - nodurile de conectare precizate prin numere (sau nume);
 - valoarea dispozitivului la care precizarea unității de măsură este opțională;
 - tipul (modelul) dispozitivului.

Deci declararea unor elemente de circuit este de forma:

rezistorul:	Rnume nod1 nod2 valoare;
condensatorul:	Cnume nod1 nod2 valoare;
tranzistorul bipolar:	Qnume colector bază emitor tipul;
sursă de tensiune:	Vnume nod+ nod- valoare;

3. precizarea tipului de analiză și a parametrilor acesteia.

2. Editarea, rulara și vizualizarea grafică a rezultatelor unui fișier PSPICE

Dupa ce programul este pornit, prin executarea **psp.bat** pe ecran apare fereastra principală Pspice Control Shell, ca în fig.1, în care se alege ca opțiune **Files** (fișiere), iar apoi **Current File** (fișier curent).




Figura 1

Circuit File Name (numele fișierului) are extensia **.cir** și poate fi scris ca în fig.2 de la tastatură sau poate fi ales unul dintre fișierele memorate anterior cu ajutorul tastei **F4**.




Figura 2

În continuare se intră în fereastra de editare selecționând **Edit** ca în fig.3 și în fereastra de editare a circuitului **Circuit Editor**, se scriu liniile succesive ale programului ca în fig.4.

Pe prima linie a programului, linie ignorată la execuție, se poate scrie titlul programului sau comentarii. Observații neexecutabile pot fi scrise și pe alte linii de program prin marcarea acestora la începutul lor cu simbolul * (asterisc).



Figura 3



Figura 4

După scrierea programului, eventualele erori de editare se pot căuta prin tastarea F6.

Salvarea se face părăsind fereastra de editare cu ESC și apoi alegând S, după cum se observă în fig. 5.




Figura 5

Etapa următoare constă în rularea programului din meniul principal, submeniul **Analysis**, comanda **Run Pspice**, după cum se observă în fig. 6.




Figura 6

În urma rulării programului rezultă un fișier de ieșire ce poate fi încărcat automat în subrutina de afișare grafică **Probe** din meniul principal.

În fig.7 este prezentat ecranul în urma execuției subrutinei de afișare grafică.




Figura 7

Reprezentarea grafică a unei mărimi se face cu ajutorul comenzii **Add_trace** (adăugare curbă), urmată de selecționarea cu **F4** a unei variabile disponibile dintr-o listă ca în fig.8, sau compunerea unei noi variabile utilizând operatori matematici.




Figura 8

Un exemplu de rezultat grafic al acestei selecții este prezentat în fig.9. Cu ajutorul cursorului **Cursor** se pot studia coordonatele punctelor graficului și deci studia comportarea circuitului.




Figura 9

3. Desfășurarea lucrării.

3.1. Urmând procedeul indicat în secțiunea 2 se editează programul următor, ce corespunde circuitului din fig. 10:

```

Amplificator limitator inversor
VCC 4 0 5V
VIN 1 0 5V
R1 1 2 10K
R2 4 3 1K
Q1 3 2 0 NPN1
.MODEL NPN1 NPN(BF=200)
.DC VIN 0 5 0.1
.END
    
```

Final CIR LAB3




Figura 10

La introducerea fiecărei linii se poate apela din fereastra de editare comanda **F3=Manual**, care are rol de ajutor și se pot obține informații suplimentare despre fiecare comandă sau descrierea dispozitivelor, inclusiv exemple. După rularea programului se va reprezenta cu ajutorul subrutinei grafice **Probe** tensiunea V(3), ce este tensiunea de ieșire a amplificatorului limitator inversor. Se vor măsura cu ajutorul cursorului valorile tensiunii de intrare pentru obținerea pragurilor de nivel logic "0" și "1" corespunzătoare unei tensiuni la ieșire V(3) de 0,5 V și respectiv 4,5V. Se vor modifica valorile axelor pentru o vizualizare cât mai corectă a mărimilor, se vor face lupe în zone de interes folosind comanda **Zoom**.

3.2. După ieșirea din fereastra grafic (**Exit Program**) se reintră în meniul principal, fig.3 și apelând **Browse Output** (căutare în conținutul fișierului de ieșire) se citește și notează conținutul fișierului de ieșire. Se reia editarea programului adăugându-se comanda **.OP** cu o linie înaintea comenzii finale **.END**. Se rulează din nou programul și se studiază din nou fișierul de ieșire apelând **Browse Output**. Se vor nota modificările care se observă.

3.3. Se modifică valoarea rezistenței R1 la valori cuprinse între 2 și 8 KΩ și se completează tabelul următor:

R1 (KΩ)	2	4	6	8
VIN pentru V(3) = 0,5V (V)				
VIN pentru V(3) = 4,5V (V)				

3.4. Se modifică valoarea factorului de amplificare în curent BF al tranzistorului Q1 la valori cuprinse între 50 și 150 și se completează tabelul următor:

BF	50	100	150
VIN pentru V(3) = 0,5V (V)			
VIN pentru V(3) = 4,5V (V)			

3.5. Se studiază caracteristica statică în polarizare directă a unei diode semiconductoare de tip 1N4148. Se cercetează mai întâi calea către biblioteca de modele. Se observă că modelul diodei conține mai mulți parametri cu valori fixate, restul având valorile implicite.

.model D1N4148 D(Is=0.1p Rs=16 CJO=2p Tt=12n Bv=100 Ibv=0.1p)

Parametrii Is și Rs sunt importanți pentru regimul de polarizare directă.

Se editează și se simulează circuitul din fig.11 observându-se curentul prin diodă.

figura 11

studiul caracteristicii statice al diodei semiconductoare

VIN 1 0 0.5V

D1 1 0 D1N4148

.LIB C:\STUDENTI\SPICE51\LIB\DIODE.LIB

.DC VIN 0 2 0.001

.END




Figura 11

Schimbați axa y, în coordonate *logaritmice* ~~liniare~~. Cum se observă efectul rezistenței serie Rs?

Puteți să demonstrați pe grafic că acesta are o valoare de 16Ω?

3.6. Se va face studiul în curent alternativ al circuitului de amplificare din fig. 12.

amplificator cu emitor comun

figura 12

VCC 4 0 8V

VS 1 0 AC 0.01V

RS 1 2 50

RB1 4 3 4K

RB2 3 0 4K

RC 4 5 2K

RE 6 0 {RVAR}

.PARAM RVAR=2K

.STEP PARAM RVAR 2K 5K 1K

RL 7 0 2K

CI 2 3 10u

CE 6 0 10u

CO 5 7 10u

Q1 5 3 6 NPN1

.MODEL NPN1 NPN(BF=200)

.AC DEC 5 10 10MEG

.END




Figura 12

În structura programului se observă comanda **.PARAM** care permite definirea rezistenței RE ca o rezistență cu valoare variabilă și comanda **.STEP** ce permite variația valorii rezistenței. De asemenea analiza de curent alternativ se face prin comanda **.AC** varianta pe decade de frecvență, cu 5 puncte pe decadă, frecvența minimă 10Hz și maximă 100MHz. Se va studia forma generală a acestor comenzi prin apelarea manualului, tasta **F3**.

Se va edita și rula acest program în Pspice și se vor reprezenta grafic amplificarea în tensiune $V(7)/V(1)$ și cea în curent $I(RL)/I(RS)$.

Folosind cursorul se vor măsura valorile acestor amplificări la diferite frecvențe și se va completa tabelul următor:

Frecvența (Hz)	10	10^2	10^3	10^4	10^5	10^6
$A_v = V(7)/V(1)$						
$A_i = I(RL)/I(RS)$						

4. Întrebări

- 4.1. Ce efect are modificarea rezistenței R1 la punctul 3.3. asupra valorii tensiunii de intrare la care se obțin pragurilor de nivel logic "0" și "1" pentru tensiunea de ieșire.
- 4.2. Ce efect are modificarea factorului de amplificare în curent BF al tranzistorului Q1 la punctul 3.4. asupra valorii tensiunii de intrare la care se obțin pragurilor de nivel logic "0" și "1" pentru tensiunea de ieșire.
- 4.3. Ce elemente din circuitul din fig.12 determină scăderea amplificărilor de tensiune și curent la frecvențe joase.